



以RC電路為例—— 如何將Arduino應用 在物理學教學上

文 | 王恒

國外近二三十年的物理學教育研究（Physics Education Research, PER）一再表明，以主動學習為核心思想設計的教學方法，其教學效果往往遠超傳統教學法。在傳統教學法中，課堂以教師為中心，學生被動學習，教師“灌輸”知識給學生，學生的學習成效很低。^[1]Interactive Lecture Demonstrations（ILD）和RealTime Physics是目前國外比較流行的主動學習教學方法，詳細做法可閱讀“參考文獻”2和3。無論是ILD還是RealTimePhysics，都需要在課堂或實驗室進行實驗，實驗數據通過傳感器獲得並即時顯示在電腦屏幕上。本澳學校一般要購買價格高昂的專業教學實驗儀器，如PASCO公司的運動傳感器、溫度傳感器等並配合相應的數據收集和處理軟件才能做到這一點。本文提供了另一種極低成本的辦法：使用Arduino控制傳感

器並收集測量數據，數據通過USB連接線實時傳回電腦軟件Excel中顯示。

Arduino是一種開源的微控制器，價格低廉，而且有極其龐大的社群支持。使用Arduino，我們可以輕易控制許多不同的傳感器並應用在物理學教學上，^[4-6]這些傳感器的價格也十分便宜，讀者可自行到淘寶網搜索。使用過Arduino的讀者都會知道，Arduino收集的數據可以通過Arduino和電腦的USB連接線實時傳回電腦並顯示在Arduino軟件的Serial Monitor上。為了讓學生更好地觀看實時收集的測量數據，我們使用一個小程序PLX-DAQ來做到Arduino和Excel的數據互傳。通過PLX-DAQ，Arduino收集的數據可以實時、按格式地傳入Excel並顯示出來，達到了使用PASCO產品的類似效果。^[7]關於PLX-DAQ的下載和使

用方法，讀者可閱讀“參考文獻”8。以下舉一例以說明如何使用上述的軟硬件於物理學教學。

我們知道電容的充電和放電需要時間，對於一個RC電路，電容的充電放電時間和電阻R和電容C的大小有關。例如，當RC電路中的電容電壓為零時對其充電，我們需要 $\tau = RC$ 秒才能將電容充電達到電源電壓的63.2%。要顯示這一動態過程，我們可以使用Arduino來做到。^[9]如圖1所示的RC電路中， $R=200k\Omega$ ， $C=100\mu F$ ，可知RC電路的時間常數 $\tau = RC=20s$ 。圖1中的D7接Arduino數位腳位7，可以通過Arduino控制此處為5V或0V（分別對應數位1或0），這樣就能夠用程序控制對RC電路中的電容進行充電或放電。A0接Arduino類比腳位0，用於測量電容電壓 U_c ；A5接Arduino類比腳位5，用於測量另一邊電阻處的電壓（實際上等於D7，此處模仿了“參考文獻”9的做法）。圖2是筆者連接的實物圖，圖3是筆者繪畫的實物示意圖，讀者可據此連接自己的電路。其中注意到我使用了電解電容，實際上此處沒有必要使用有極性的電容。

連接好電路後，編寫如下的Arduino程序，程序中已經附加了注釋說明代碼的意義。代碼中的許多行“Serial.print”和“Serial.println”

語句都是用於將Arduino的數據通過PLX-DAQ傳入Excel，要明白它們的意義就必須閱讀“參考文獻”8。




圖 1




圖 2



圖 3



```
float R = 0.2; //定義電阻的大小·以兆歐作單位·用作計算電流
int charge; //此變量用作判斷是否要給RC電路電壓
int chargePin = 7; //定義對RC電路給5V電壓的腳位(對於Arduino的GND腳位來說)

void setup() {
    pinMode(chargePin, OUTPUT); //設定上述腳位為輸出模式

    //設定Arduino與電腦的串行通訊速度·測量數據將通過連接Arduino與電腦的USB線傳入電腦
    Serial.begin(9600);

    //以下是Arduino與Excel附加工具PLX DAQ的通信所用到的語句
    Serial.println("CLEARDATA"); //清除Excel中的原有數據
    Serial.println("LABEL,Time,Voltage,Current"); //設定Excel第一列各欄的名稱
    Serial.println("RESETTIMER"); //重置PLX DAQ的時間計數
    //設定PLX DAQ中那三個checkbox的名字·本程序實際上只用到第一個checkbox
    Serial.println("CUSTOMBOX1,LABEL,Charge"); //此checkbox控制是否給電路電壓
    Serial.println("CUSTOMBOX2,LABEL,N/A"); //沒用
    Serial.println("CUSTOMBOX3,LABEL,N/A"); //沒用
    Serial.println("CUSTOMBOX1,SET,0"); //設定第一個checkbox的原始狀態為沒有選中
}

void loop() {
    Serial.println("CUSTOMBOX1,GET"); //獲取第一個checkbox的選擇狀態
    charge = Serial.readStringUntil(10).toInt(); //使變量charge等於獲取的結果(1或0)
    //通過變量charge判斷對RC電路施加5V還是0V電壓
    if(charge){
        digitalWrite(chargePin, HIGH);
    } else {
        digitalWrite(chargePin, LOW);
    }

    //將A0腳位和A5腳位的電壓讀出,這個讀數的範圍為0~1023·對應0~5V電壓
    int V_A0 = analogRead(0);
    int V_A5 = analogRead(5);
    //將A0和A5腳位讀出的數據轉為0~5V的形式
    float V_C = (V_A0)*5.0/1023.0;
    float V = (V_A5)*5.0/1023.0;
    //使用歐姆定律計算電路電流·因為電阻單位為兆歐·所以電流的單位為微安
    float I = (V - V_C) / R;

    //以下是Arduino與Excel附加工具PLX DAQ的通信所用到的語句
    Serial.print("DATA,TIMER,"); //輸出數據到Excel·TIMER指時間
    Serial.print(V_C, 3); //輸出電容電壓到Excel
    Serial.print(","); //PLX DAQ輸出數據所用到的格式
    Serial.println(I, 3); //輸出電路電流到Excel
    //延遲一秒·以免讀出太多數據
    delay(1000);
}
```

對上述代碼進行編釋並上傳到Arduino，打開PLX-DAQ的Excel檔案，按下PLX-DAQ介面的“Connect”按鈕，就會見到通過Arduino測量和計算的電容電壓和電路電流出現在Excel表格中。通過PLX-DAQ介面名為“Charge”的checkbox（請看代碼中的注釋），我們可以控制D7數位腳位的電壓，達到對電容充電或放電的目的。配合Excel的作圖功能，我們可以將這些實時收集的測量數據同步以線圖的形式顯示出來，如圖4所示。教師在課前備好課後，就可以直接使用這個PLX-DAQ的Excel文檔和Arduino去上課，所有的數據和圖表的顯示格式都已經設定好。根據圖4的結果，在 $\Delta t=23.6-3.13=20.5s \approx \tau$ 的時間間隔中，電容電壓由0V增加至3.19V，達到了電源電壓5V的63.8%，和理論結果符合得很好。

總結來說，本文以顯示RC電路中電容的充放電數據為例，說明了如何將Arduino的測量數據實時輸出到Excel中。運用Arduino和Excel，教師可以以極低的成本，進行大量不同的課堂演示實驗，甚至可以開展自己的ILD或RealTime Physics教學。🌱




圖 4

[參考文獻]

1. R. Thornton and D. Sokoloff (1998). Assessing student learning of Newton's laws: the force and motion conceptual evaluation and the evaluation of active learning laboratory and lecture curricula, *Am. J. Phys.* **66**, 338.
2. D. Sokoloff and R. Thornton (1997). Using Interactive Lecture Demonstrations to create an active learning environment, *Phys. Teach.* **35**, 340.
3. D. Sokoloff, P. Laws, and R. Thornton (2007). RealTime Physics: active learning labs transforming the introductory laboratory, *Eur. J. Phys.* **28**, S83.
4. F. Bouquet, J. Bobroff, M. Fuchs-Gallezot, and L. Maurines (2017). Project-based physics labs using low-cost open-source hardware, *Am. J. Phys.* **85**, 216.
5. M. McCaughey (2017). An Arduino-based magnetometer, *Phys. Teach.* **55**, 274.



6. C. Galeriu, S. Edwards, and G. Esper (2017). An Arduino investigation of Simple Harmonic Motion, Phys. Teach. **52**, 157.
7. D. Nichols (2017). Arduino-based data acquisition into Excel, LabVIEW, and MATLAB, Phys. Teach. **55**, 226.
8. NetDevil, PLX-DAQ version 2, <https://forum.arduino.cc/index.php?topic=437398.0>
9. C. Galeriu, C. Letson, and G. Esper(2015). An Arduino investigation of the RC Circuit, Phys. Teach. **53**, 285.

王恒

澳門坊眾學校教師

教學人員培訓活動



為提升本澳教學人員專業水準，有效地應對教育工作的各種挑戰，本局將持續安排各類教學人員的培訓活動，讓教學人員按照需要參加。

為使教學人員及早預備及規劃其專業發展，有關教學人員培訓活動詳情及報名時間，請留意本局向學校派發的活動章程或瀏覽本局網站。

- 聯絡：教育資源中心
- 地址：澳門南灣大馬路926號
- 電話：8395 9299 / 8395 9200

- 傳真：8395 9131
- 網址：www.dsej.gov.mo/cre
- 電郵：cre@dsej.gov.mo